
      12 25≡2↑1↓⎕TS  
0 
      ⎕←now←⎕TS  
2019 9 16 10 7 49 959 
      xmas←2019 12 25 0 0 0 0 
      xmas-now 
0 3 9 ¯10 ¯7 ¯49 ¯959 
      ⍝ 
      +2 ⎕NQ #'DateToIDN' now 
43723 
      dn←{+2 ⎕NQ #'DateToIDN'⍵} 
      dn¨xmas now 
43823 43723 
      -/dn¨xmas now 
100 
      (24 60 60 1000⊥3↓now)÷86400000 
0.4221060069 
      dn←{(2⎕NQ#'DateToIDN'⍵)+(24 60 60 1000⊥3↓⍵)÷86400000} 
      dn¨xmas now 
43823 43723.42211 
      -/dn¨xmas now 
99.57789399 
      ⍝ 
      2000 0 0 ¯1 ¯1 ¯1 ¯1 +? 100 12 28 24 60 60 1000 
2099 3 14 22 55 34 576 
      d1←{2000 0 0 ¯1 ¯1 ¯1 ¯1 +? 100 12 28 24 60 60 1000}¨⍳1000 
      ]disp 2↑d1 
┌→────────────────────┬─────────────────────┐ 
│2046 8 22 19 12 2 770│2022 4 9 18 51 25 150│ 
└~───────────────────→┴~───────────────────→┘ 
      ⎕SIZE 'd1' 
56032 
      d2←dn¨d1 
      2↑d2 
53560.80003 44659.78571 
      ⎕SIZE 'd2' 
8032 
      (⎕SIZE 'd2')÷⎕SIZE 'd1' 
0.1433466591 
      ⍝ 
      1 ¯1⎕DT xmas now 
43823 43723.42211 
      1⎕DT xmas now 
43823 43723.42211 
      -/1⎕DT xmas now 
99.57789399 
      ⍝ 
      1⎕DT⊂now 
43723.42211 
      ¯1⎕DT 1⎕DT⊂now 
 2019 9 16 10 7 49 959  
      ¯1⎕DT 250 + 1⎕DT⊂now 
 2020 5 23 10 7 49 959  
      ⍝ 
      ¯1 ⎕DT 0 
 1899 12 31 0 0 0 0  
      ¯1 ⎕DT 1 
 1900 1 1 0 0 0 0  
      ¯1 ⎕DT ¯0.5 
 1899 12 30 12 0 0 0  
      ¯1 ⎕DT ¯100000 
 1626 3 17 0 0 0 0  
      ⍝ 
      t←'cf.dcf' ⎕FCREATE 0 
      'Hello' ⎕FAPPEND t 
      ⎕FREAD t 1 
Hello 



      ⎕FRDCI t 1 
136 0 9.411749349E10 
      ⎕←cfts←3⊃⎕FRDCI t 1 
9.411749349E10 
      ¯1 2⎕DT cfts 
 2019 9 16 9 8 11 433  
      1 2⎕DT cfts 
43723.38069 
      ⍝ 
      ⍝ 3: UNIX time 
      ⍝ 4-9: Julian time variants 
      4 ⎕DT ⊂now 
2458742.922 
      ⍝ 10: Microsoft Win32 FILETIME 
      ⍝ 11: Microsoft CLR DateTime.ticks 
      10 ⎕DT ⊂now 
1.321310207E17 
      ⍝ 12-13: Excel 
      12 ⎕DT⊂1900 2 28 
59 
      12 ⎕DT⊂1900 3 1 
61 
      12 ⎕DT⊂1900 2 29 
DOMAIN ERROR: Invalid timestamp 
      12 ⎕DT⊂1900 2 29 
         ∧ 
      ⍝ 14: Microsoft OLE Automation Date 
      ⍝ 
      ¯1 ⎕DT 'J' 
 2019 9 16 10 8 24 732  
      ¯1 ⎕DT 'Z' 
 2019 9 16 9 8 25 450  
      1 ⎕DT 'JZ' 
43723.42252 43723.38086 
      24×-/1 ⎕DT 'JZ' 
1 
      ¯1 ⎕DT 'C' 
 2019 9 16 12 8 27 927  
      ⎕←h←1 ⎕DT 'C' 
43723.50593 
      ⍝ 
      dtf←1200⌶ 
      'DD/MM/YYYY hh:mm:ss' dtf h 
 16/09/2019 12:08:32  
      'D/M/YY hh:mm:ss' dtf h 
 16/9/19 12:08:32  
      '_D/_M/YYYY hh:mm:ss' dtf h 
 16/ 9/2019 12:08:32  
      'D MMM YYYY hh:mm:ss' dtf h 
 16 SEP 2019 12:08:32  
      'D MMMM YYYY hh:mm:ss' dtf h 
 16 SEPTEMBER 2019 12:08:32  
      'D Mmmm YYYY hh:mm:ss' dtf h 
 16 September 2019 12:08:32  
      'M _M MM Mmm Mmmm' dtf h 
 9  9 09 Sep September  
      'WWWW-"W"ww-d' dtf h 
 2019-W38-1  
      'WWWW-"W"ww-d' dtf 1 ⎕DT ⊂2019 12 31 
 2020-W01-2  
      'd Ddd' dtf h  
 1 Mon  
      'Dddd D Mmmm YYYY hh:mm:ss' dtf h 
 Monday 16 September 2019 12:08:32  
      'Dddd Doo Mmmm YYYY hh:mm:ss' dtf h 
 Monday 16th September 2019 12:08:32  
      'Dddd Mmmm D YYYY t:mm:sspp' dtf h 



 Monday September 16 2019 12:08:32pm  
      '__fi__Dddd Doo mmmm YYYY hh:mm:ss' dtf h 
 Maanantai 16. syyskuu 2019 12:08:32  
      dtf←1200⌶⍠'Language' 'fi' 
      'Dddd Doo mmmm YYYY hh:mm:ss' dtf h 
 Maanantai 16. syyskuu 2019 12:08:32  
      '__en__Dddd Doo mmmm YYYY hh:mm:ss' dtf h 
 Monday 16th September 2019 12:08:32  
      '__da__Dddd Doo mmmm YYYY hh:mm:ss' dtf h 
 Mandag 16. september 2019 12:08:32  
      '__fr__Dddd Doo mmmm YYYY hh:mm:ss' dtf h 
 Lundi 16 septembre 2019 12:08:32  
      '__ru__Dddd Doo mmmm YYYY hh:mm:ss' dtf h 
 Понедельник 16 Сентябрь 2019 12:08:32  
      '__cy__Dddd Doo mmmm YYYY hh:mm:ss' dtf h 
DOMAIN ERROR: Unknown language 'cy' 
      '__cy__Dddd Doo mmmm YYYY hh:mm:ss'dtf h 
                                         ∧ 
      ⊃⎕NGET'Dict.json' 
{                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                    
  "cy": {                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                            
    "MonthNames": [                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                  
      "Ionawr", "Chwefror", "Mawrth", "Ebrill", "Mai", "Mehefin",                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                    
      "Gorffennaf", "Awst", "Medi", "Hydref", "Tachwedd", "Rhagfyr"                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                  
    ],                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                               
    "ShortMonthNames": [                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                             
      "Ion", "Chw", "Maw", "Ebr", "Mai", "Meh",                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                      
      "Gor", "Awst", "Medi", "Hyd", "Tach", "Rhag"                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                   
    ],                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                               
    "WeekdayNames": [                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                
      "Dydd Sul", "Dydd Llun", "Dydd Mawrth", "Dydd Mercher",                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                        
      "Dydd Iau", "Dydd Gwener", "Dydd Sadwrn"                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                       
    ],                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                               
    "ShortWeekdayNames": [                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                           
      "Sul", "Llun", "Maw", "Mer", "Iau", "Gwen", "Sad"                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                              
    ],                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                               
    "MorningAfternoon": [                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                            
      "yb", "yh"                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                     
    ],                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                               
    "Ordinals": [                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                    
      "af", "il", "ydd", "ydd", "ed", "ed", "fed", "fed", "fed", "fed", "eg",                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                        
      "fed", "eg", "eg", "fed", "eg", "eg", "fed", "eg", "fed", "ain", "ain",                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                        
      "ain", "ain", "ain", "ain", "ain", "ain", "ain", "ain", "ain"                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                  
    ]                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                
  }                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                  
}                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                                    
 
      dict←⎕JSON⊃⎕NGET'Dict.json' 
      dict.cy.ShortWeekdayNames 
 Sul  Llun  Maw  Mer  Iau  Gwen  Sad  
      dtf←1200⌶⍠('Dictionary'dict)('Language' 'cy') 
      'Dddd Doo mmmm YYYY hh:mm:ss' dtf h 
 Dydd Llun 16eg Medi 2019 12:08:32  
      ⍝ 
      presents←'onsdag 25 december 2019 07:30:00' 
      ⎕USING←'System' 'System.Globalization,System.Globalization.dll' 'Dyalog' 
      lang←CultureInfo.CreateSpecificCulture⊂'da-DK' 
      format←'dddd dd MMMM yyyy HH:mm:ss' 
      ret←⎕NEW ByRef 
      DateTime.TryParseExact presents format lang DateTimeStyles.None ret 
1 
      ret.Value.ToOADate 
43824.3125 
      ¯1 14 ⎕DT ret.Value.ToOADate 
 2019 12 25 7 30 0 0  
      ⍝ 
      ⍝ Happy Christmas! 


